MCF example for luamplib(LuaLaTeX)

Author : Akira Yamaji Date : January 22, 2022
Located at : http://www.ctan.org/pkg/mcf2graph

1 MCF example

use molecular data base file 'mcf_exa_data.mcf'
`FM(fm)` : molecular formula (calculated)
`MW(mw)` : molecular weight (calculated)

1.1 Erythromycin

(pass mcf to MC())

```
beginfigm("EN:Erythromycin","MW:733.93")
  fsize:=(120mm,30mm);
  MC(
    <30,#1,<-120,60,60,60,-60,60,60,-60,60,60,60,-60,60,60,##,&1,
    14:O,13:/*Et,{1,9}://O,{2',4,6^-35,8,10',12^35}:/*_,
    {6^35,11,12^-35}:*/OH,
    @$3,\*,O,30~zb,|,?6‘.7,6:O,#.5,{5~wf,3^35}:/_,4:/*OH,3^-35:/*O!,##,
    @$5,\*^30‘1.7,O,!~zb,|,?6‘.7,6:O,#.5,5:/*_,2:*/OH,3:/*N?!
  )
  ext(defaultscale:=0.8;
      label.lrt("fm: "&cal_FM,(0,h-5mm));
      label.lrt("mw: "&cal_MW,(0,h-9mm));
      label.lrt("MW: "&inf_MW,(0,h-13mm));)
endfigm;
```

fm: C37H67NO13
mw: 733.9267
MW: 733.93



1.2 Paclitaxel

(pass mcf to beginfigm())

```
beginfigm("EN:Paclitaxel","MW:853.918",
  %-------------------------------------------------------------------
  ": ?6,5=dl,@3,#1,36,45,45,45,45,##,&5,-4=?6,-4=?4,-1=wb,-3=wf,-1:O, ",
  ": 4:??,6:/_,{3^-60,15}:*/OH,8:/*H^-60,9:*/_^60,10://O,             ",
  ": @1,\,O,!,//O,!,*/OH,!,/Ph,60~wf,NH,-60,//O,60,Ph,                ",
  ": @7,\*,O,-45,//O,60,Ph,11:*/OCO!>rl,12:/*OCO!^-15>lr              ")
  %-------------------------------------------------------------------
  fsize:=(140mm,30mm); if check(mc)=0: MCat(0,0.5)(scantokens(mc)) fi
  sw_numbering:=Atom;  if check(mc)=0: MCat(0.6,0.5)(scantokens(mc)) fi
  sw_numbering:=Bond;  if check(mc)=0: MCat(1,0.5)(scantokens(mc)) fi
endfigm
```







1.3 Chlorophyll a

(read data-base file)

```
beginfigm("f:mcf_data_base.mcf","t:EN","v:Chlorophyll a","NO:-")
  sw_output:=Fig+Calc+Mcode;                  %%%% output temp-mc.aux %%%%
  fsize:=(100mm,30mm);
  if check(mc)=0:
    MC(scantokens(mc))
    VerbatimTeX("\gdef\EN{"&inf_EN&"}\gdef\MW{"&inf_MW&"}");
    VerbatimTeX("\gdef\mw{"&cal_MW&"}\gdef\fm{"&cal_FM&"}");
  fi
endfigm
\end{mplibcode}
\verbatiminput{temp-mc.aux}             %%%% input temp-mc.aux %%%%
{\tt ** EN:\EN \quad mw:\MW \quad MW:\mw \quad fm:\fm}%
```




```
<-36,#1,?5,@3,\,54,?5,@-2,\,54,?5,@-2,\,54,?5,@-2,\,&5,@6,22,70,&8,##,
 {4,6,8,10,14,16,18,21,23,27}=dl,@4,\‘1.48~vf,Mg,&17~vb,@11,&27,@27,&23,
 {4,11,17,23}:N,{1~zf,9,15,21}:/_,14:/!,20:/!!,25:/*COO!,26://O,
 @2,*\^-6,!2,//O,!,O,!2,!!,|,!13,{1,5,9,13}:/_

** EN:Chlorophyll a  mw:893.509  MW:893.4889  fm:C55H72MgN4O5
```

1.4 Dinophysistoxin-1

(read data-base file + pass mcf to beginfigm())

```
beginfigm("t:EN","v:Okadaic acid","EN:Dinophysistoxin-1",
          "MW:819",":,38:*/_,65=red")  %%%% add methyl group (color red) %%%%
  sw_output:=Fig+Calc+Mcode;           %%%% output temp-mc.aux %%%%
  fsize:=(120mm,20mm);
  if check(mc)=0: MC(scantokens(mc))
    VerbatimTeX("\gdef\EN{"&inf_EN&"}\gdef\MW{"&inf_MW&"}");
    VerbatimTeX("\gdef\mw{"&cal_MW&"}\gdef\fm{"&cal_FM&"}");
  fi
endfigm;
\end{mplibcode}
\verbatiminput{temp-mc.aux}             %%%% input temp-mc.aux %%%%
{\tt ** EN:\EN \quad mw:\MW \quad MW:\mw \quad fm:\fm}%
```




```
<30,?6,@4,?6,@-4,\,!3,<-12,?5,@-3,<-12,?6,-3=?6,@-3,*\,!3,
 ?6,@-4,?6,@6,\,!,/*_^-40,*/OH^20,!,//O,!1,OH,
 3=wb,11=dl,15=dr,17=wf,19=wf,38=wb,{5,7,16,24,25,33,42}:O,
 32:*/H^60,10:/_,{12,31,37’}:*/_,27://_,28:/OH,{3,29}:/*OH
,38:*/_,65=red

** EN:Dinophysistoxin-1  mw:819  MW:819.0294  fm:C45H70O13
```

1.5 Maitotoxin

(read data-base file)

```
%-------------------------------------------------------------------------------
\begin{mplibcode}
  beginfigm("t:EN","v:Maitotoxin")
    sw_output:=Fig+Calc+Mcode;          %%%% output temp-mc.aux %%%%
    fsize:=(120mm,60mm); fmargin:=(0,3mm); sw_frame:=Outside;
    if check(mc)=0: MC(scantokens(mc))
      VerbatimTeX("\gdef\EN{"&inf_EN&"}\gdef\MW{"&inf_MW&"}");
      VerbatimTeX("\gdef\mw{"&cal_MW&"}\gdef\fm{"&cal_FM&"}");
    fi
  endfigm
\end{mplibcode}
\verbatiminput{temp-mc.aux}             %%%% input temp-mc.aux %%%%
{\tt ** EN:\EN \quad mw:\MW \quad MW:\mw \quad fm:\fm}%
%-------------------------------------------------------------------------------
```



```
<55.8,?6,-4=?7 ,{-4,-3,-3,-3}=?6,@-3,\,!3,?6,{-4,-3,-3,-3}=?6,@-3,\,?6,-3=?6,
 @-3,\,!3,60,<-30,?6,-3=?6,@-3,30,<30,?6,{-3,-3}=?6,-3=?7,{-4,-3,-3}=?6,
 @-2,\,?6,-3=?6,-3=?7,{-3,-3}=?6,-3=?8,-3=dl,{-5,-3,-3,-3}=?6,
 {5,7,15,16,23,24,32,40,41,48,49,58,59,72,73,82,83,90,91,99,
   100,107,113,114,122,123,130,131,140,141,148,149}:O,
 {1^60,2,26,28,29,51,54,61,63,68,75^60,78,109}:*/OH,
 {11,20,35,45,52,55,65,69,86}:/*OH,
 {3,8,13,17,21,33,38,42,56,70,84,92,101,106,111,128,138,142,146,150}:/*H^-60,
 {4,14,22,34,39,43,47',57',71',81,89,98,102,116,121,125,129,133}:*/H^60,
 {6,46,50,53,60,67,74}:*/H^-60,
 {9,18,85,93,112,139,143,147}:*/_^60`1,{80,88,97,108',115,120,124}:/*_^-60`1,
 @$6,\,|,!11,60~dr,-60,60,OH,{2',7,10}:*/OH,{1,3,8'}:*/_,11://_,12:/_,
   @6,\,O,30,SOO,30,"O{Na}",
 @$36,-45~zf,O,30,SOO,30,"O{Na}",
 @$150,\,|,!7,{1,2}:/*OH,4:*/_,5:/*_,7=dl

** EN:Maitotoxin  mw:3425.86  MW:3425.856  fm:C164H256Na2O68S2
```

1.6 TCA cycle




```
beginfigmy"EN:TCA cycle")
fsize:=(160mm,75mm);
max_blength:=5mm;
COOH:='(//O,!,OH);
HOCO:='(OH,!,//O,);
MCat(0.33,   1)(<30,HOCO,!,//O,!2,COOH)                    % Oxaloacetate
MCat(0.66,   1)(<30,HOCO,!4,COOH,@-4‘1,\,COOH,4:/OH^-165)  % Citrate
MCat(1,      1)(<30,HOCO,!2,!~dr,!,COOH,@-4‘1,\,COOH)      % cis-Aconitate
MCat(1,   0.58)(<30,HOCO,!4,COOH,@-4,\‘1,COOH,5:/OH)       % Isocitrate
MCat(1,   0.05)(<30,HOCO,!3,//O,!,COOH,@-4,\‘1,COOH)       % Oxalosuccinate
MCat(0.66,0.05)(<30,HOCO,!3,//O,!,COOH)                    % alfa-Ketoglutarate
MCat(0.33,0.05)(<30,HOCO,!3,//O,!,"{S-CoA}")               % Succinyl-CoA
MCat(0,   0.05)(<30,HOCO,!3,COOH)                          % Succinate
MCat(0,   0.55)(<30,HOCO,!,!~dr,!,COOH)                    % Fumarate
MCat(0,      1)(<30,HOCO,!3,COOH,3:/OH)                    % L-Malate
ext(
defaultfont:="uhvr8r";
defaultscale:=0.75;
ext_setup;
save dx; pair dx; dx:=(12mm,0);
label.bot("Oxaloacetate",p1+dx);   label.bot("Citrate",p2+dx);
label.bot("cis-Aconitate",p3+dx);  label.bot("Isocitrate",p4+dx);
label.bot("Oxalosuccinate",p5+dx); label.bot("alfa-Ketoglutarate",p6+dx);
label.bot("Succinyl-CoA",p7+dx);   label.bot("Succinate",p8+dx);
label.bot("Fumarate",p9+dx);       label.bot("L-Malate",p10+dx);
sw_label_emu:=1;
ext_setup;
r_arrow(10mm)(  0)(p1+ ( 1.1w1, 0.3h1))("Acetyl-CoA",1.5)(" CoA-SH",1);
r_arrow(10mm)(  0)(p2+ ( 1.1w2, 0.4h2))("",0)("H2O",1);
r_arrow( 8mm)(270)(p3+ ( 0.5w3,-0.4h3))("H2O",1)("",0);
r_arrow( 8mm)(270)(p4+ ( 0.5w4,-0.4h4))("NAD+",1)("NADH2+",1);
r_arrow(10mm)(180)(p5+ (-0.1w5, 0.4h5))("",0)("CO_2_",1);
r_arrow(10mm)(180)(p6+ (-0.1w6, 0.5h6))("NAD+,CoA-SH",1.7)("NADH2+,CO2",1);
r_arrow(10mm)(180)(p7+ (-0.1w7, 0.5h7))("GDP,Pi",1.7)("GTP,CoA-SH",1);
r_arrow( 8mm)( 90)(p8+ ( 0.4w8, 1.2h8))("FAD",1)("FADH2",1);
r_arrow( 8mm)( 90)(p9+ ( 0.4w9, 1.2h9))("H2O",1)("",0);
r_arrow(10mm)(  0)(p10+( 1.1w10,0.3h10))("NAD+",1)("NADH2+",1.5);
defaultscale:=1.5;
label("TCA-cycle",(0.5w,0.5h));
)
endfigm
```